RPG Committee

1st Experience with Procedures

Definitions:
pr
Prototype.  Basically a PList for procedures.  These are D specs and most resemble data structures.  One way of think about this is that a Prototype defines a procedure.
pi
Procedure Interface.  Basically a *entry PList for procedures.
· If named, it must be the same name as the procedure (or program name if mainline).  **Note: A mainline “pi” must be named.

· Not needed if a procedure has no parms and no return value.
· Must look like and follow a prototype of the same name if a procedure does have parms.
· *Note #1:  Once a Prototype is defined (for instance – in the mainline), it does not need to be defined.

· *Note #2:  The Procedure Interface must look like the Prototype, but the field names can be different.

· *Note #3: Prototype “field names” are really “descriptors”.  They have no function in the program.  They can technically be blank.

· The Procedure Interface can be defined for use in a return value.
extpgm
Keyword used on a Prototype.  Specifies external program to call.
value
Keyword used on a Prototype field.  Specifies that the field value rather than address (aka referencce) be passed.
· *Note#1: This protects the fields in the calling procedure as the called procedure now has its own field addresses.  (Stated another way:  The called procedure can modify the values for its own processing purposes, but the modifications will not be reflected in the calling procedure.)
· *Note #2: Per IBM manual, this cannot be used on prototypes using the “extpgm” keyword.

const
Keyword used on a Prototype field.  Specifies that the field passed is read only (cannot be changed).  However, it is still passed by reference (address).  The IBM manual had an attention box on this:
· If you are using a Prototype in the called procedure, the compiler will check to ensure that the field is not changed (aka if you change the value in the called procedure, the compile will fail).
· If the called program is not using a Prototype, be sure that the called program does not change the value.  It does not mention consequences.

options
Keyword used on a Prototype field.  The following options can be specified:
· *nopass:  When this is used, the field does not need to be passed.  When this is used, all subsequent fields on the prototype must also have “*nopass”.

· *Note:  This is really something for the Prototype compile checker.  In use, it is very similar to just not passing the parameter using PList and checking %parms.

· *omit:  The text “*omit” is used if the parameter is not being passed.  It is the responsibility of the called procedure to check %addr on the given parrm to see if it was passed.
· *varsize:  This allows variable length fields to be passed.  It is the responsibility of the called program to correctly process this (either by using %len or using another parm to determine length).
· *string: This is used to pass the address of a string.  When using the Prototype (via CallP), a string or an address of a string can be used.  The data type of the field with “*string” as a keyword looks like it must be a pointer.
· *rightadj: This right adjusts the value in the Prototype.

Some things that I discovered on my “journey” (coding):

· Everything defined in the mainline is accessible to the procedures in that program.

· Things defined in a procedure are not accessible by the mainline or other procedures unless they are either in a bound module and “export” is used or they are in a service program and “export” is used.  *Note: an “import” must be used in the calling procedure to access fields made available by “export”.
· Keeping *Inzsr may be a “good thing” to define KLists.

· The “value” keyword cannot be used in the “pi” of a mainline.

· The “return” opcode can be used in a procedure to exit just that procedure.
· *Note:  The “$Exit” subroutine in screen programs that I have grown fond of will no longer work.  You can no longer set on *InLR and Return to get out of the program.
· Use the “value” keyword to protect fields in the calling procedure used as parms.  This prevents the called program from modifying them.  Please note:  The called procedure can modify the values it receives, but the changes will not be sent back to the calling procedure.
· The names on the Prototype in the calling procedure, the Prototype in the called procedure, and the Procedure Interface can all be different, but the definition (length, type, etc) must match.
· A Procedure return value may be specified.

· Specify the definition on the “pr” and “pi” spec line.  “n” for Boolean and “*” for pointer are valid as well as other standard field types.
· A “return” opcode must be used and a return value must be specified in factor 2.  (**Note:  This is only if a return value is specified.)
Some things that I discovered on my “journey” (running):

· Procedures show in the job stack when active.

· An error encountered in a procedure issues an error message.  If that error message results in that procedure canceling, then the calling procedure gets an error message also.
· Native debug will not “step into” a procedure.  A break point must be specified.

Example from my project:
Things to note are in blue.  Here is a summary:

1) The mainline Prototype (pr) is defined before the Procedure Interface (pi)

2) The procedure Interface (pi) definition matches its Prototype (pr) and has the same name.

3) The procedure Interface (pi) is named the same as the program.

4) Every procedure has a Prototype.

5) $Proc_MstID:

a. The Prototype (pr) is defined in the mainline.  It needs to be so that the mainline can use $Proc_MstID.
b. The Procedure Interface (pi) is defined in the procedure.

c. The Procedure Interface (pi) is named the same as the procedure.  (**Note:  A “pi” does not need to be named unless it is a mainline “pi”.)
6) $LogError:

a. A procedure does not need a Procedure Interface (pi) if there are no –

i. Parameters

ii. Nor a return value.

b. A Prototype (pr) still exists in the mainline.

7) Fields from the mainline including file fields.  Not depicted, but please note:  File access (aka “CHAIN” and “READ”) can be done in procedures on files defined in the mainline.

8) The field defined in the procedure $Proc_MstID is not used in the mainline or the other procedure.

      **********************************************************************************************

      *¹GO7871: Red Bull: Determine Master IDs to process.

      **********************************************************************************************

     H  Option(*srcstmt : *nodebugio : *showcpy : *expdds)

     H  debug(*yes)

      **********************************************************************************************

     F*   DATE   *  PROGRAMMER           *  MODIFICATION

      **********************************************************************************************

     F* 09/06/05 *  Craig R. Lockhart    * RN-5570. Creation.

      **********************************************************************************************

      *¹Files.

      * Distributor ID master.

     FMSDISTMx3 IF   E           K Disk

      **********************************************************************************************

      *¹Stand-alone fields

      **********************************************************************************************

     D Dft_Dt          s               d   inz(d'0001-01-01')

     D Today_Dt        s               d   inz(*sys)

     D Wk_Start_Dt     s               d   inz(d'0001-01-01')

     D Wk_End_Dt       s               d   inz(d'0001-01-01')

     D Wk_Rcpient      s                   like(MSRCPIENT)

     D                                     inz('RED BULL')

      * Paramter Master Distributor ID.

     D Pm_MstID        s                   like(MSMSTDSTID)

      * Paramter Distributor ID.

     D Pm_DstId        s                   like(MSDSTID)

      * Paramter error message.

     D Pm_ErrMsg       s             80

     D Sv_MstID        s                   like(MSMSTDSTID)

     D I_1st           s               n

      **********************************************************************************************

      *¹Constants.

      **********************************************************************************************

      **********************************************************************************************

      *¹Arrays.

      **********************************************************************************************

      **********************************************************************************************

      *¹Data structures.

      **********************************************************************************************

      **********************************************************************************************

      *¹ProtoTypes for procedures.

      **********************************************************************************************

      *.............................................................................................

      * THIS program.

      * ProtoType for *Entry parms.

     D GO7871          pr

     D  In_MstID                           like(MSMSTDSTID)

     D  In_StartDate                 10

     D  In_EndDate                   10

      * *Entry parms.

     D GO7871          pi

     D  In_MstID                           like(MSMSTDSTID)

     D  In_StartDate                 10

     D  In_EndDate                   10

      *.............................................................................................

      * To process a master ID.

     D $Proc_MstID     pr

     D  WkP_MstID                          like(MSMSTDSTID)

     D                                     value

     D  WkP_StartDate                10    value

     D  WkP_EndDate                  10    value

      *.............................................................................................

      * Log errors.

     D $LogError       pr

      **********************************************************************************************

      *¹Main routine.

      **********************************************************************************************

      *.............................................................................................

      * Master distributor ID was given.  Process only it.

     C                   Select

     C                   When          In_MstID <> *Blanks

     C     Ky_MSDISTMx3  Chain     MSDISTMx3

      * Process if Master Distributor Id exists.

     C                   If            %found(MSDISTMx3)

     C                   CallP     $Proc_MstID(In_MstID:In_StartDate:In_EndDate)

      * Log error (for report and Email).

     C                   Else

     C                   Eval      Pm_ErrMsg = 'Specified Master ID is not -

     C                             setup.'

     C                   CallP     $LogError

     C                   Endif

      *.............................................................................................

      * Master distributor ID was not given.  Loop through all Master IDs for Red Bull.

     C                   Other

     C                   Eval      I_1st = *on

     C     Wk_Rcpient    Setll     MSDISTMx3

     C     Wk_Rcpient    ReadE     MSDISTMx3

     C                   Dow           not %eof(MSDISTMx3)

     C                   If            I_1st

     C                             or  MSMSTDSTID <> Sv_MstID

     C                   CallP     $Proc_MstID(MSMSTDSTID:In_StartDate:

     C                             In_EndDate)

     C                   Eval      I_1st = *off

     C                   Eval      Sv_MstID = MSMSTDSTID

     C                   Endif

     C     Wk_Rcpient    ReadE     MSDISTMx3

     C                   Enddo

      * If no Distributor IDs exist for Red Bull, log error (for report and Email).

     C                   If            I_1st

     C                   Eval      Pm_errMsg = 'No Distributor IDs set up for -

     C                             Red Bull!!!'

     C                   CallP     $LogError

     C                   Endif

     C                   Endsl

      *.............................................................................................

      * Exit program.

     C                   Eval      *InLR = *on

     C                   Return

      **********************************************************************************************

      *¹*InzSR - Initialization subroutine.

      **********************************************************************************************

     C     *Inzsr        Begsr

     ¹*.............................................................................................

      * **Note:  The *entry parameters are in a procedure interface (pi).  See prototype D specs.

     ¹*.............................................................................................

      * Key lists.

      * Distributor Id master (MSDISTMx3).

     C     Ky_MSDISTMx3  KList

     C                   KFld                    Wk_Rcpient

     C                   KFld                    In_MstID

     ¹*.............................................................................................

      * Initial processing.

     C                   Endsr

      **********************************************************************************************

      *¹$Proc_MstID - Process a given master distributor ID.

      **********************************************************************************************

     P $Proc_MstID     B

      *.............................................................................................

      * Procedure interface.

     D $Proc_MstID     pi

     D  Wk_MstID                           like(MSMSTDSTID)

     D                                     value

     D  Wk_StartDate                 10    value

     D  Wk_EndDate                   10    value

      *.............................................................................................

      * Procedure fields.

     D Wk_Update       s              1

      * ............................................................................................

      * Is this a regular process (as opposed to a re-run)?  If so, set update flag to update

      * process date.

     C                   If            In_MstID     = *blanks

     C                             and In_StartDate = *blanks

     C                             and In_EndDate   = *blanks

     C                   Eval      Wk_Update = 'Y'

     C                   Else

     C                   Eval      Wk_Update = 'N'

     C                   Endif

      * ............................................................................................

      * Determine start date.

     C                   If            Wk_StartDate = *blanks

     C     *iso          Move      MSLPROC_Dt    Wk_StartDate

     C                   Endif

      * ............................................................................................

      * Determine end date.

     C                   If            Wk_EndDate = *blanks

     C     Today_Dt      SubDur    1:*d          Wk_End_Dt

     C     *iso          Move      Wk_End_Dt     Wk_EndDate

     C                   Endif

      * ............................................................................................

      * Start date can never be greater that End date.

     C     *iso          Move      Wk_StartDate  Wk_Start_Dt

     C     *iso          Move      Wk_EndDate    Wk_End_Dt

     C                   If            Wk_Start_Dt > Wk_End_Dt

     C                   Eval      Wk_StartDate = Wk_EndDate

     C                   Endif

      * ............................................................................................

      * Okay folks, we are ready to process this Master ID ...

     C                   Call      'GO7871CL'

     C                   Parm                    Wk_MstID

     C                   Parm                    Wk_StartDate

     C                   Parm                    Wk_EndDate

     C                   Parm                    Wk_Update

     P                 E

      **********************************************************************************************

      *¹$LogError - Log error.

      **********************************************************************************************

     P $LogError       B

      * ............................................................................................

      * Log error.

     C                   Eval      Pm_MstID = In_MstID

     C                   Eval      Pm_DstID = *blanks

     C                   Call      'GO7878'

     C                   Parm                    Pm_MstID

     C                   Parm                    Pm_DstID

     C                   Parm                    Pm_ErrMsg

     P                 E
Example of “extpgm”, “const” and “*nopass”:
 *  Delete User Space                                                                          

D qusdltus        pr                  extpgm('QUSDLTUS')                                       

D  Pr2_US_Name                        like(APIUS_Name)                     User space nam & lib

D                                     const                                                    

D  pr2_Error                      *   options(*nopass)                     Error information.  
Example of return value from procedure:
Note that the procedure name can have an extended name, just like fields can.

D isParentChildValid...            

D                 pr              n
Note that:

1) The procedure return value is used as part of the conditional statement.

2) The “()” after the procedure.  These are not required, but serve as a visual reference that this is a procedure.

C                   If        isParentChildValid()             

C                   Eval      p@ReturnCode = RETURN_CODE_TRUE  

C                   Elseif    isIgnored                        

C                   Eval      p@ReturnCode = RETURN_CODE_IGNORE

C                   Else                                       

C                   Eval      p@ReturnCode = RETURN_CODE_FALSE 

C                   Endif                                      

Note that:

1) The Procedure Interface (pi) is defined (colored red in this case to help it stand out).  This defines the return value.

2) A field is defined with the same definition and the “pi”.

3) A “return” opcode is used and the defined field is used in factor 2.  This returns a value the calling procedure.

P isParentChildValid...                       

P                 b                           

D                 pi              n           

D returnValue     s               n   Inz(*On)
{… more code …}

C                   Return    returnValue
P                 e                      

List of RPGLE programs that use procedures:
Whse cost: GSP9000 and GSP9100 in library MASVUSRGO.  This has some really cool aspects of /copy using /define.

Free form: SQLRPGLE: GSC2101 in MAXH.  This also highlight some innovative uses of SQL.

General: GSC2100, GSC225, and GSC2231 in GOLIBR.

